

JOINDRE.TEXTE

Définition



La fonction **JOINDRE.TEXTE** permet de **combiner le texte** de **plusieurs cellules** dans **une seule cellule** en incluant un **séparateur** entre **chaque texte à combiner**.

NB : **JOINDRE.TEXTE** imbrique souvent la fonction **TRIER**, **FILTRE** et/ou **SI** dans les arguments **Texte**. C'est une **version améliorée** de la fonction **CONCAT**.

Syntaxe

JOINDRE.TEXTE(**Séparateur**;**Ignorer_vider**;Texte1;**[Texte2]**;...)

Argument	Description
Séparateur (obligatoire)	Séparateur entre chaque Texte . Cet élément peut être vide ou contenir un ou plusieurs caractères entre guillemets (""). Si le Séparateur est un nombre , alors il sera traité comme du texte .
Ignorer_vider (obligatoire)	Ignorer ou inclure les cellules vides : ⇒ VRAI : ignorer les cellules vides . ⇒ FAUX : inclure les cellules vides .
Texte1 (obligatoire)	1^{er} élément de texte à joindre . Cet élément peut être une cellule (A3) , une plage de cellules (A1:A8) , ou un élément texte entre guillemets ("").
[Texte2] (facultatif)	Tu peux spécifier jusqu'à 252 éléments de texte de cette façon.

Exemple 1 : comprendre la fonction JOINDRE.TEXTE

Ici, tu veux **combiner le texte** des colonnes **A**, **B** et **C** (colonne **D**) en les **séparant** d'une **barre oblique** " / ". Tu veux **inclure** les **cellules vides**.

	A	B	C	D
1	Adresse	Code postal	Ville	Combiner le texte
2	10 Rue Maurice Barrès	57000	METZ	10 Rue Maurice Barrès / 57000 / METZ (1) =JOINDRE.TEXTE(" / ";FAUX;A2:C2)
3	23 Rue Cavenne		LYON	(2) 23 Rue Cavenne / / LYON =JOINDRE.TEXTE(" / ";FAUX;A3:C3)

Remarques

- ⇒ (1) Tu dois mettre le **Séparateur** " / " pour l'**insérer** entre **chaque texte**.
- ⇒ (2) **VRAI** (**ignorer** les **cellules vides**) aurait **supprimé** le **2^e Séparateur** " / ".
- ⇒ Les cellules **D2 à D3** contiennent une **formule** qui est **liée** aux cellules **A2 à C3**. Si tu **modifies** le contenu des cellules **A2 à C3**, alors les cellules **D2 à D3** seront **impactées**. Afin de **convertir la formule en texte** dans les cellules **D2 à D3**, tu dois les **copier**, puis effectuer un **collage spécial** dans les cellules **D2 à D3** en **collant uniquement** les **"Valeurs"**.

Exemple 2 : TRIER imbriquée dans la fonction JOINDRE.TEXTE

Ici, tu veux **lister tous les fruits par ordre alphabétique** (cellule **A7**) en les **séparant** d'une **virgule** ", ". Tu veux **ignorer** les **cellules vides**.

	A	B
1	Fruits	Quantité disponible
2	Pommes	5 000
3	Framboises	10 000
4	Bananes	1 000
5		
6	Fraises	7 000
7	Bananes, Fraises, Framboises, Pommes (1) =JOINDRE.TEXTE(", ";VRAI;TRIER(A2:A6))	

Remarques

- ⇒ (1) Si tu veux mettre une **virgule** entre **chaque texte**, alors tu dois la mettre entre **guillemets** suivie d'un **espace** dans le **Séparateur** (" , ").
- ⇒ (1) **VRAI** permet **d'ignorer** les **cellules vides**.
- ⇒ (1) **TRIER** permet de **trier automatiquement les fruits de A à Z** (**par ordre alphabétique**).

Exemple 3 : FILTRE imbriquée dans la fonction JOINDRE.TEXTE

Ici, tu veux **lister tous les fruits ayant une quantité disponible d'au moins 5 000** (cellule **B7**) en les **séparant** d'une **virgule** ", ". Tu veux **ignorer** les **cellules vides**.

	A	B
1	Fruits	Quantité disponible
2	Pommes	5 000
3	Framboises	10 000
4	Bananes	1 000
5		
6	Fraises	7 000
7	Pommes, Framboises, Fraises (1) =JOINDRE.TEXTE(", ";VRAI;FILTRE(A2:A6;B2:B6>=5000))	

Remarques

- ⇒ (1) Si tu veux mettre une **virgule** entre **chaque texte**, alors tu dois la mettre entre **guillemets** suivie d'un **espace** dans le **Séparateur** (" , ").
- ⇒ (1) **FILTRE** permet de **filtrer automatiquement** les **fruits** ayant une **quantité >= 5 000**.

Exemple 4 : TRIER et FILTRE imbriquées dans la fonction JOINDRE.TEXTE

Ici, tu veux **lister tous les fruits par ordre alphabétique ayant une quantité disponible d'au moins 5 000** (cellule **B7**) en les **séparant** d'une **virgule** ", ". Tu veux **ignorer** les **cellules vides**.

	A	B
1	Fruits	Quantité disponible
2	Pommes	5 000
3	Framboises	10 000
4	Bananes	1 000
5		
6	Fraises	7 000
7		Fraises, Framboises, Pommes (1) =JOINDRE.TEXTE(", ";VRAI;TRIER(FILTRE(A2:A6;B2:B6>=5000)))

Remarques

- ⇒ (1) Si tu veux mettre une **virgule** entre **chaque texte**, alors tu dois la mettre entre **guillemets** suivie d'un **espace** dans le **Séparateur** (" , ").
- ⇒ (1) **TRIER** permet de **trier automatiquement les fruits de A à Z (par ordre alphabétique)**.
- ⇒ (1) **FILTRE** permet de **filtrer automatiquement** les **fruits** ayant une **quantité >= 5 000**.

Exemple 5 : SI imbriquée dans la fonction JOINDRE.TEXTE

Ici, tu veux **lister les joueurs du PSG** (cellule **D2**) en les **séparant** d'une **virgule** ", ". Tu veux **ignorer** les **cellules vides**.

	A	B	C	D
1	Joueur	Équipe		Joueurs du PSG
2	K. MBAPPÉ	PSG		K. MBAPPÉ, L. MESSI, NEYMAR Jr. (1) =JOINDRE.TEXTE(", ";VRAI;SI(B2:B6="PSG";A2:A6;""))
3	L. MESSI	PSG		
4	D. PAYET	OM		
5	S. MANDANDA	OM		
6	NEYMAR Jr.	PSG		

Remarques

- ⇒ (1) **Cette formule** est une **alternative** à la fonction **FILTRE imbriquée dans JOINDRE.TEXTE**.
- ⇒ (1) **Lecture de la formule** : **combiner les joueurs** (A2:A6) en les **séparant** d'une **virgule** (" , ") **si** les **joueurs** jouent au **PSG** (B2:B6="PSG") **sinon** afficher **rien** ("").
- ⇒ (1) **SI** permet de **spécifier des critères de concaténation** pour la fonction **JOINDRE.TEXTE**.